math16502.mw
Maple 10 Worksheet for Problems in Math 165 - Calculus for Business.

First restart and load plots and student:

```
> restart: with( student):with (plots):
```

Section 1.1 Problem 65 p.13 (Edition 8/e)
The demand function p = D(x) is
D(x) = - .02 x + 29
i.e. a price p = - .02 x + 29 will produce a consumer demand for x units
and the to total cost function C(x) is
C(x) = 1.43 x^2 + 18.3 x + 15.6.
Later we will try to justify the model.
Here 15.6 may be thought of as fixed cost - before any units are produced and
x (1.43 x + 18.43) the incremental cost of producing the first x units.
The revenue function R(x) = (number of unit) x (price per unit) is R(x) = x D(x).
The profit function is P(x) - C(x).

Set up functions. The author (JL) prefers to write functions as `procedures."

```
> Demand:= proc(x)
    description `produces a function of x`:
    - .02*x + 29:
  end proc;
```

$$Demand := \textbf{proc}(x)\ \textbf{description}\ \text{`}\textit{produces a function of x}\text{`};\ -0.02 * x + 29\ \textbf{end proc} \quad (1)$$

```
> Cost:= proc(x)
     description `[Total] Cost function`:
     1.43 * x^2 + 18.3 * x + 15.6;
  end proc;
```

$$Cost := \textbf{proc}(x)\ \textbf{description}\ \text{`}\textit{[Total] Cost function}\text{`};\ 1.43 * x\text{^}2 + 18.3 * x + 15.6\ \textbf{end proc} \quad (2)$$

```
> Revenue:= proc(x)
      description `[Total] Revenue function`;
      x*Demand(x):
  end proc;
```

$$Revenue := \textbf{proc}(x)\ \textbf{description}\ \text{`}\textit{[Total] Revenue function}\text{`};\ x * Demand(x)\ \textbf{end proc} \quad (3)$$

```
> Profit:= proc(x)
      description `[Total] Profit`:
      Revenue(x) - Cost(x):
  end proc;
```

$$Profit := \textbf{proc}(x)\ \textbf{description}\ \text{`}\textit{[Total] Profit}\text{`};\ Revenue(x) - Cost(x)\ \textbf{end proc} \quad (4)$$

```
> simplify(Revenue(x));
  simplify(Cost(x));
  simplify(Profit(x));
```

$$-0.02000000000\, x\, (x - 1450.)$$

$$1.430000000\, x^2 + 18.30000000\, x + 15.60000000$$

$$-1.450000000\,x^2 + 10.70000000\,x - 15.60000000 \quad (5)$$

Plot Revenue and Cost on the same graph.

```
> plot([Revenue(x),Cost(x)],x=0..10, y = 0..400,color=[red,blue],
  thickness = [1,3],legend=[`Revenue(x)`,`Cost(x)`]);
```



Note that Cost(x) > Revenue(x) only on a narow range.
Next graph Profit(x).

```
> plot(Profit(x),x=0..10,thickness=3,legend=`Profit(x)`);
```



It appears that Profit is POS only on a narrow range.

```
> solve(Profit(x) > 0,x);
```

$RealRange(Open(2.), Open(5.379310345))$ (6)

Maple [or your graphing calculator!] can find the point of maximum profit.

```
> with(Optimization);
  Maximize(Profit(x),initialpoint={x = 3});
```

[*ImportMPS*, *Interactive*, *LPSolve*, *LSSolve*, *Maximize*, *Minimize*, *NLPSolve*, *QPSolve*]

$[4.139655172414, [x = 3.68965517241379314]]$ (7)

Optimizing Parabolas.

Since Profit (x) has a very special form:
P(x) = A x^2 + B x + C,
The **roots** of P(x) = 0 are found at
root\pm = (- B \pm sqrt(B^2 - 4 A C))/(2 A),
and the **vertex** (MAX if A < 0, MIN if A > 0) is at
x = -B/(2A).
PARABOLAS [ONLY] are SPECIAL: Note that the vertex is located at the MIDPOINT of the segment between the two roots.

```
> Quad:= proc(A,B,C,x)
```

```
      description `quadratic A x^2 + B x + C`;
      A*x^2 + B*x + C;
   end proc;
>
```

Quad := **proc**(*A*, *B*, *C*, *x*) (8)

description `*quadratic A x^2 + B x + C*`;

$A * x^2 + B * x + C$

end proc

```
> solve(Quad(A,B,C,x)=0,x);
```

$$-\frac{1}{2}\ \frac{B-\sqrt{B^2-4\,A\,C}}{A},\ -\frac{1}{2}\ \frac{B+\sqrt{B^2-4\,A\,C}}{A} \qquad (9)$$

Now look at the quadratic Profit(x) = -1.45 x^2 -10.7 x -15.6

```
> A:=-1.45;B:=10.7;C:=-15.6;
```

$$A := -1.45$$

$$B := 10.7$$

$$C := -15.6 \qquad (10)$$

Find where P(x) > 0 by finding the roots of P(x) = 0.

```
> solve(Quad(A,B,C,x)=0,x);
```

$$2.,\ 5.379310345 \qquad (11)$$

Find the Maximum Value at the vertex:

```
> vertex:= -B/(2*A);Max_Profit:=Quad(A,B,C,vertex);
```

$$vertex := 3.689655172$$

$$4.13965517 \qquad (12)$$

```
>
```